



EBook Gratis

APRENDIZAJE excel

Free unaffiliated eBook created from
Stack Overflow contributors.

#excel

Tabla de contenido

Acerca de.....	1
Capítulo 1: Empezando con Excel.....	2
Observaciones.....	2
Examples.....	2
Ejemplos de algunas fórmulas básicas disponibles en Excel.....	2
Capítulo 2: Contando celdas únicas.....	5
Examples.....	5
Usando COUNTIF ().....	5
Usando FREQUENCY () y MATCH ().....	5
Capítulo 3: Conversión de base.....	6
Introducción.....	6
Observaciones.....	6
Examples.....	6
Conversión de base.....	6
Capítulo 4: Crear conexión a otros archivos de Excel.....	7
Examples.....	7
Creando una conexión a otros archivos de Excel para consultar.....	7
Capítulo 5: Excelencias y límites.....	8
Observaciones.....	8
Examples.....	14
Especificaciones de excel.....	14
Capítulo 6: Excelentes prácticas.....	15
Examples.....	15
Utilizar tablas de Excel.....	15
Utilizar tablas de pivote.....	15
Extraer datos de tablas dinámicas con GetPivotData.....	15
Mejores prácticas.....	15
Capítulo 7: Excelentes redondeos y precisión.....	17
Introducción.....	17
Sintaxis.....	17

Parámetros.....	17
Observaciones.....	18
Examples.....	18
Usando la función ROUND.....	18
Usando las funciones TRUNC & INT.....	19
Usando la función MROUND.....	19
Usando las funciones CEILING & FLOOR.....	20
Usando la función FIJA.....	20
Capítulo 8: Formato de celda.....	22
Introducción.....	22
Observaciones.....	22
Examples.....	22
Formato de número.....	22
Capítulo 9: Fórmulas de matriz.....	24
Examples.....	24
Suma de Producto de Rangos.....	24
Capítulo 10: Función DATEDIF.....	26
Sintaxis.....	26
Parámetros.....	26
Observaciones.....	26
Examples.....	26
Periodo de conteo entre fechas.....	26
Capítulo 11: Función MATCH.....	28
Introducción.....	28
Parámetros.....	28
Observaciones.....	28
Examples.....	29
Comprobando si una dirección de correo electrónico aparece en una lista de direcciones.....	29
Combinando MATCH con INDEX.....	30
Capítulo 12: Función SUMPRODUCT.....	31
Introducción.....	31

Sintaxis.....	31
Observaciones.....	31
Examples.....	31
Usando SUMPRODUCT con rangos numéricos.....	31
Usando SUMPRODUCT con matrices booleanas.....	31
Capítulo 13: Índice de coincidencia para Excel.....	33
Introducción.....	33
Examples.....	33
Índice de coincidencia vertical.....	33
Coincidencia de índice horizontal.....	33
Capítulo 14: Un proceso fácil para convertir datos mensuales a trimestrales en Excel.....	34
Introducción.....	34
Examples.....	34
Una forma fácil de realizar una tarea rutinaria a veces tediosa:.....	34
Capítulo 15: VLOOKUP.....	38
Introducción.....	38
Sintaxis.....	38
Parámetros.....	38
Observaciones.....	38
Examples.....	39
Usar VLOOKUP para obtener el apellido de una persona de su ID de empleado.....	39
Usar VLOOKUP para calcular el porcentaje de bonificación (ejemplo con el comportamiento "p.....	39
Utilizando VLOOKUP con coincidencia aproximada.....	40
Usando VLOOKUP con coincidencia exacta.....	40
Creditos.....	42

Acerca de

You can share this PDF with anyone you feel could benefit from it, downloaded the latest version from: [excel](#)

It is an unofficial and free excel ebook created for educational purposes. All the content is extracted from [Stack Overflow Documentation](#), which is written by many hardworking individuals at Stack Overflow. It is neither affiliated with Stack Overflow nor official excel.

The content is released under Creative Commons BY-SA, and the list of contributors to each chapter are provided in the credits section at the end of this book. Images may be copyright of their respective owners unless otherwise specified. All trademarks and registered trademarks are the property of their respective company owners.

Use the content presented in this book at your own risk; it is not guaranteed to be correct nor accurate, please send your feedback and corrections to info@zzzprojects.com

Capítulo 1: Empezando con Excel

Observaciones

Microsoft Excel es una hoja de cálculo de alta utilidad para uso en cálculo y programación de macros. Excel también cuenta con compatibilidad con Visual Basic para aplicaciones. Los datos se organizan y evalúan en una cuadrícula de celdas, donde se pueden formular o mostrar en gráficos.

Examples

Ejemplos de algunas fórmulas básicas disponibles en Excel

Excel tiene muchas fórmulas integradas. A continuación, se incluyen algunos ejemplos de algunas de las fórmulas básicas que puede ser útil saber al comenzar con Excel:

Nota importante: ¡El nombre y la sintaxis de estas fórmulas varían en el idioma de su instalación de Excel! Por ejemplo esta función aquí:

en inglés: `=left("Hello World",5)`

Lo mismo en alemán: `=links("Hello World";5)`

¡No solo el nombre es diferente, sino también algunas partes menores de la sintaxis! Por ejemplo en la versión alemana ; Se utiliza en lugar de , para separar los parámetros.

Todas las fórmulas se pueden ingresar en cualquier celda escribiendo primero "=" y luego el nombre de la fórmula. Puede escribir las fórmulas directamente en la celda o puede seleccionar la celda y escribir la fórmula en la barra de fórmulas. Para mostrar todas las fórmulas en una hoja, puede seleccionar el botón "Mostrar fórmulas" en el grupo de botones "Auditoría de fórmulas" en la pestaña Fórmulas:



Una gran cantidad de fórmulas utilizan un rango de células. Para dar un rango, debe hacer referencia a la primera celda en la esquina superior izquierda, como `A1` y la esquina inferior derecha, como `B3` y colocar dos puntos ":" entre ellos como este `A1:B3`. Esto le dará un rango de 6 celdas en dos columnas y tres filas:

	A	B
1		
2		
3		
4		

SUMA()

En las celdas D1 - D3 están los valores 2,3 y 6 En la celda D4 tenemos la siguiente fórmula:
`=SUM(D1:D3)` , el resultado es 11, que se muestra en la celda.

`Sum` Agrega todos los números en un rango de celdas.

Si tiene esa celda seleccionada, la Fórmula se mostrará en la barra de Fórmula sobre la cuadrícula de celdas:

=SUM(D1:D3)			
D	E	F	
2			
3			
6			
11			

COUNT () / COUNTA () / COUNTIF () / COUNTBLANK ()

En las celdas E1 - E10 están los valores 2,3,5,6, blanco, 8,9, blanco, 11 y 12.

En la celda F2 está la fórmula: `=count (E1:E10)` , el resultado es 8 que se muestra en la celda.

En la celda F3 está la fórmula: `=counta (E1:E10)` , el resultado es 8 que se muestra en la celda.

En la celda F4 está la fórmula: `=countif (E1:E10, ">5")` , el resultado es 5, que se muestra en la celda.

En la celda F5 está la Fórmula: `=countblank (E1:E10)` , el resultado es 2 que se muestra en la celda.

`Count ()` Cuenta el número de celdas en un rango que contiene números.

`Counta ()` Cuenta el número de celdas en un rango que no están vacías.

`CountIF ()` Cuenta el número de celdas en un rango que cumplen con una condición dada.

`CountBlank ()` Cuenta el número de celdas vacías en un rango específico de celdas.

	E	F
2		
3		=COUNT(E1:E10)
5		=COUNTA(E1:E10)
6		=COUNTIF(E1:E10,">5")
		=COUNTBLANK(E1:E10)
8		
9		
11		
12		

También hay muchas fórmulas de cadena.

IZQUIERDA () / DERECHA () / MID ()

En la celda D1 está "Hello World!".

En la celda E1 está la fórmula: =left (D1, 5) , el resultado es: "Hola"

En la celda F1 está la fórmula: =right (D1, 6) , el resultado es: "¡Mundo!"

En la celda G1 está la fórmula: =mid (D1, 7, 5) , el resultado es: "Mundo"

left Devuelve el número especificado de caracteres desde el inicio de una cadena de texto.
=LEFT(String, Number of Characters)

right Devuelve el número especificado de caracteres desde el final de una cadena de texto.
=RIGHT(String, Number of Characters)

mid Devuelve los caracteres desde la mitad de una cadena de texto, dada la posición inicial y la longitud. =MID(String, Start Position, Number of Characters)

Estas tres fórmulas cuentan los caracteres de la cadena que comienzan con la posición 1 siendo el primer carácter.

D	E	F	G
Hello World!	=LEFT(D1,5)	=RIGHT(D1,6)	=MID(D1,7,5)

D	E	F	G
Hello World!	Hello	World!	World

Lea Empezando con Excel en línea: <https://riptutorial.com/es/excel/topic/906/empezando-con-excel>

Capítulo 2: Contando celdas únicas

Examples

Usando COUNTIF ()

```
=SUMPRODUCT((A1:A100<>"")/COUNTIF(A1:A100,A1:A100&""))
```

cuenta *valores de celda únicos* dentro de A1: A100, *excluyendo* celdas en blanco y unidades con una cadena vacía ("").

¿Como hace eso? Ejemplo:

```
A1:A100 = [1, 1, 2, "apple", "peach", "apple", "", "", -, -, -, ...]
```

Se necesita agregar & "" a la matriz para convertir las celdas en blanco (-) en cadenas vacías ("").
Resultado:

```
A1:A100&"" = ["1", "1", "2", "apple", "peach", "apple", "", "", "", "", "", ...]
```

Después de este truco, se puede aplicar COUNTIF (). Ambos "" y - se cuentan como iguales:

```
COUNTIF(A1:A100,A1:A100&"") = [2, 2, 1, 2, 1, 2, 94, 94, 94, 94, 94, ...]
```

Para obtener el recuento de todas las celdas únicas, excluyendo espacios en blanco y "", podemos dividir

```
(A1:A100<>""), which is [1, 1, 1, 1, 1, 1, 0, 0, 0, 0, 0, ...]
```

por nuestro resultado intermedio, COUNTIF (A1:A100,A1:A100&"") , y la suma de los valores.

```
SUMPRODUCT((A1:A100<>"")/COUNTIF(A1:A100,A1:A100&""))  
= (1/2 + 1/2 + 1/1 + 1/2 + 1/1 + 1/2 + 0/94 + 0/94 + 0/94 + 0/94 + 0/94 + ...)  
= 4
```

Usando FREQUENCY () y MATCH ()

```
=SUMPRODUCT(IF(FREQUENCY(MATCH(A1:A100,A1:A100,0),MATCH(A1:A100,A1:A100,0))>0,1))
```

Lea *Contando celdas únicas en línea*: <https://riptutorial.com/es/excel/topic/6263/contando-celdas-unicas>

Capítulo 3: Conversión de base

Introducción

Excel proporciona funciones que lo ayudarán a convertir de decimal a binario, octal y hexadecimal, y viceversa.

Observaciones

Tenga en cuenta que no hay 0 o 0x en las funciones.

Examples

Conversión de base

Con el número 100 en la celda A1, los resultados de estos cálculos.

```
=DEC2BIN (A1)  
=DEC2OCT (A1)  
=DEC2HEX (A1)  
=BIN2DEC (A1)  
=OCT2DEC (A1)  
=HEX2DEC (A1)
```

es

```
1100100  
144  
64  
4  
64  
256
```

tenga en cuenta que las primeras 3 funciones están justificadas a la izquierda, ya que son cadenas, y las últimas 3 están justificadas a la derecha, ya que son numéricas.

Lea **Conversión de base en línea**: <https://riptutorial.com/es/excel/topic/10708/conversion-de-base>

Capítulo 4: Crear conexión a otros archivos de Excel

Examples

Creando una conexión a otros archivos de Excel para consultar

Hay muchas formas excelentes de hacer esto, que otros ya han sugerido. Siguiendo a lo largo de "obtener datos de Excel a través de la pista SQL", aquí hay algunos consejos.

Excel tiene el "Asistente de conexión de datos" que le permite importar o vincular desde otra fuente de datos o incluso dentro del mismo archivo de Excel. Como parte de Microsoft Office (y los sistemas operativos) hay dos proveedores de interés: el antiguo "Microsoft.Jet.OLEDB" y el último "Microsoft.ACE.OLEDB". Búsquelos al configurar una conexión (como con el Asistente de conexión de datos). Una vez conectado a un libro de Excel, una hoja de cálculo o un rango es el equivalente de una tabla o vista. El nombre de la tabla de una hoja de trabajo es el nombre de la hoja de trabajo con un signo de dólar ("\$\$") que se le anexa y está rodeado de corchetes ("[" y "]"); de un rango, es simplemente el nombre del rango. Para especificar un rango de celdas sin nombre como su fuente de registros, agregue la notación estándar de fila / columna de Excel al final del nombre de la hoja entre corchetes.

Nombrar el rango de los datos que desea consultar utilizando el administrador de nombres es muy útil, ya que denominarlo "Base de datos" puede tratar los datos en una hoja de cálculo como una tabla de base de datos, permitiéndole ejecutar declaraciones SQL en sus datos para un fácil acceso. También vale la pena señalar que el rango nombrado no debe tener datos en blanco o faltantes, ya que causará que el SQL se rompa.

El SQL nativo será (más o menos) el SQL de Microsoft Access. (En el pasado, se llamaba JET SQL; sin embargo, Access SQL ha evolucionado y creo que JET está obsoleto como tecnología anterior). Ejemplo, leyendo una hoja de trabajo: `SELECT * FROM [Sheet1 $]` Ejemplo, leyendo un rango: `SELECT * FROM MyRange` Por ejemplo, leyendo un rango de celdas sin nombre: `SELECCIONAR * DE [Hoja1 $ A1: B10]` Hay muchos, muchos libros y sitios web disponibles para ayudarlo a trabajar con los detalles.

Lea [Crear conexión a otros archivos de Excel en línea](https://riptutorial.com/es/excel/topic/6328/crear-conexion-a-otros-archivos-de-excel):

<https://riptutorial.com/es/excel/topic/6328/crear-conexion-a-otros-archivos-de-excel>

Capítulo 5: Excelencias y límites

Observaciones

Especificaciones y límites de Excel ([Excel 2016-2013](#) , [Excel 2010](#) , [Excel 2007](#))

Especificaciones y límites de la hoja de trabajo y del libro de trabajo

Característica	Límite máximo
Libros abiertos	Limitado por la memoria disponible y los recursos del sistema
Tamaño de la hoja de trabajo	1,048,576 filas por 16,384 columnas
Ancho de columna	255 caracteres
Altura de la fila	409 puntos
Saltos de página	1.026 horizontales y verticales.
Número total de caracteres que puede contener una celda	32,767 caracteres
Caracteres en un encabezado o pie de página	255
Número máximo de líneas de alimentación por celda	253
Hojas en un cuaderno de ejercicios	Limitado por la memoria disponible (el valor predeterminado es 1 hoja)
Colores en un libro de trabajo	16 millones de colores (32 bits con acceso completo al espectro de color de 24 bits)
Vistas nombradas en un libro de trabajo	Limitado por la memoria disponible

Característica	Límite máximo
Formatos de celda únicos / estilos de celda	64,000
Llenar estilos	256
Línea de peso y estilos	256
Tipos de fuente únicos	1.024 fuentes globales disponibles para su uso; 512 por libro
Formatos de números en un libro de trabajo	Entre 200 y 250, dependiendo de la versión de idioma de Excel que haya instalado.
Nombres en un libro de trabajo	Limitado por la memoria disponible
Windows en un libro de trabajo	Limitado por la memoria disponible
Hipervínculos en una hoja de trabajo	66.530 hipervínculos
Paneles en una ventana	4
Hojas vinculadas	Limitado por la memoria disponible
Escenarios	Limitado por la memoria disponible; un informe resumido muestra solo los primeros 251 escenarios
Cambio de celdas en un escenario.	32
Celdas ajustables en Solver	200
Funciones personalizadas	Limitado por la memoria disponible
Rango de zoom	10 por ciento a 400 por ciento
Informes	Limitado por la memoria disponible
Ordenar referencias	64 en una sola clase; ilimitado cuando se utilizan ordenaciones secuenciales

Característica	Límite máximo
Deshacer niveles	100
Campos en un formulario de datos	32
Parámetros del libro de trabajo	255 parámetros por libro de trabajo
Elementos mostrados en las listas desplegables de filtros	10,000
Celdas no contiguas que pueden ser seleccionadas	2.147.483.648 células
Límites máximos de almacenamiento de memoria y tamaño de archivo para libros de modelo de datos	<p>El entorno de 32 bits está sujeto a 2 gigabytes (GB) de espacio de direcciones virtuales, compartido por Excel, el libro de trabajo y los complementos que se ejecutan en el mismo proceso. La parte del espacio de direcciones de un modelo de datos puede ejecutar hasta 500 - 700 megabytes (MB), pero podría ser menor si se cargan otros modelos de datos y complementos. El entorno de 64 bits no impone límites estrictos en el tamaño del archivo. El tamaño del libro de trabajo está limitado solo por la memoria disponible y los recursos del sistema. Excel 2016 ofrece la funcionalidad Large Address Aware que permite que Excel 2016 de 32 bits consuma el doble de la memoria cuando los usuarios trabajan en un sistema operativo Windows de 64 bits. Para obtener más información, consulte Cambio de capacidad de Large Address Aware para Excel. Nota: Agregar tablas al modelo de datos aumenta el tamaño del archivo. Si no planea crear relaciones complejas de modelo de datos utilizando muchos orígenes de datos y tipos de datos en su libro, desmarque la casilla Agregar estos datos al modelo de datos cuando importe o cree tablas, tablas dinámicas o conexiones de datos.</p>

Especificaciones y límites de cálculo.

Característica	Límite máximo
Precisión del número	15 dígitos
Número negativo permitido más pequeño	-2.23E-308
Número positivo permitido más	2.23E-308

Característica	Límite máximo
pequeño	
Mayor número positivo permitido	1.00E + 308
El mayor número negativo permitido	-1.00E + 308
El mayor número positivo permitido a través de la fórmula	1.7976931348623158e + 308
El mayor número negativo permitido a través de la fórmula	-1.7976931348623158e + 308
Longitud de los contenidos de la fórmula	8,192 caracteres
Longitud interna de la fórmula	16,384 bytes
Iteraciones	32.767
Arreglos de hojas de trabajo	Limitado por la memoria disponible
Rangos seleccionados	2,048
Argumentos en una función	255
Niveles anidados de funciones.	64
Categorías de funciones definidas por el usuario	255
Número de funciones de hoja de trabajo disponibles	341
Tamaño de la pila de operandos	1,024
Dependencia entre hojas de trabajo	64,000 hojas de trabajo que pueden referirse a otras hojas
Dependencia de fórmula de matriz entre hojas de trabajo	Limitado por la memoria disponible
Dependencia del área	Limitado por la memoria disponible
Dependencia de área por hoja de trabajo	Limitado por la memoria disponible
Dependencia en una sola celda	4 mil millones de fórmulas que pueden depender de una sola celda.

Característica	Límite máximo
Longitud del contenido de la celda vinculada de libros cerrados	32.767
Fecha más temprana permitida para el cálculo	1 de enero de 1900 (1 de enero de 1904, si se utiliza el sistema de fechas de 1904)
Última fecha permitida para el cálculo	31 de diciembre de 9999
La mayor cantidad de tiempo que se puede introducir	9999: 59: 59

Especificaciones y límites de gráficos

Característica	Límite máximo
Gráficos vinculados a una hoja de trabajo	Limitado por la memoria disponible
Hojas de trabajo referidas por un gráfico	255
Series de datos en un gráfico	255
Puntos de datos en una serie de datos para gráficos 2-D	Limitado por la memoria disponible
Puntos de datos en una serie de datos para gráficos 3-D	Limitado por la memoria disponible
Puntos de datos para todas las series de datos en un gráfico	Limitado por la memoria disponible

Especificaciones y límites de informes de tablas y gráficos dinámicos

Característica	Límite máximo
Informes de tabla dinámica en una hoja	Limitado por la memoria disponible
Artículos únicos por campo	1,048,576
Campos de fila o columna en un informe de tabla dinámica	Limitado por la memoria disponible
Filtros de informe en un informe de tabla dinámica	256 (puede estar limitado por la memoria disponible)
Campos de valor en un informe de tabla dinámica	256

Característica	Límite máximo
Fórmulas de elementos calculados en un informe de tabla dinámica	Limitado por la memoria disponible
Filtros de informe en un informe de gráfico dinámico	256 (puede estar limitado por la memoria disponible)
Campos de valor en un informe de gráfico dinámico	256
Fórmulas de elementos calculados en un informe de gráfico dinámico	Limitado por la memoria disponible
Longitud del nombre MDX para un elemento de tabla dinámica	32.767
Longitud de una cadena de tabla dinámica relacional	32.767
Elementos mostrados en las listas desplegadas de filtros	10,000

Libros de trabajo con la configuración "Permitir cambios por más de un usuario ..." habilitada Si la configuración de Permitir cambios por más de un usuario ... está activada para un libro de trabajo, se aplica la siguiente información. Se puede acceder a esta configuración haciendo clic en la pestaña Revisar> Compartir libro de trabajo. Tenga en cuenta que en las versiones más recientes de Excel, el botón Compartir libro de trabajo se ha ocultado. Para mostrarlo, haga clic en Archivo> Opciones> Barra de herramientas de acceso rápido. Abra la lista en Elegir comandos de y seleccione Todos los comandos. Desplácese hacia abajo en esa lista hasta que vea Compartir libro de trabajo (heredado). Seleccione ese elemento y haga clic en Agregar. Haga clic en Aceptar. El botón Compartir libro ahora está en la parte superior de la ventana de Excel y tiene este aspecto: Compartir libro

Característica	Límite máximo
Usuarios que pueden abrir el archivo al mismo tiempo.	256
Vistas personales en el cuaderno de ejercicios.	Limitado por la memoria disponible
Días que se cambia la historia se mantiene.	32.767 (el valor predeterminado es 30 días)
Libros de trabajo que se pueden combinar a la vez	Limitado por la memoria disponible
Celdas que se pueden resaltar	32.767

Característica	Límite máximo
Colores utilizados para identificar los cambios realizados por diferentes usuarios cuando se activa el resaltado de cambios	32 (cada usuario se identifica con un color diferente; los cambios realizados por el usuario actual se resaltan en azul marino)
Tablas de Excel en el libro de trabajo	0 (cero). Nota: un libro de trabajo que contiene una o más tablas de Excel no puede tener habilitada la configuración Permitir cambios por más de un usuario ...

Especificaciones y límites de Excel ([Excel 2016-2013](#) , [Excel 2010](#) , [Excel 2007](#))

Examples

Especificaciones de excel

Especificaciones y límites de [Excel](#) ([Excel 2016-2013](#) , [Excel 2010](#) , [Excel 2007](#))

Lea [Excelencias y límites en línea](#): <https://riptutorial.com/es/excel/topic/10833/excelencias-y-limites>

Capítulo 6: Excelentes prácticas

Examples

Utilizar tablas de Excel

Al seleccionar una matriz y elegir "Insertar tabla" en el menú, crea una tabla que le permite extraer e insertar datos de una manera estructurada. Digamos que ha nombrado a la tabla "SalesEvents" y dado que la primera fila (encabezado) lee "Salesperson" "Date" "Sales Amount", puede calcular el monto de las ventas así: `=SUM(SalesEvents[Sales Amount])`. La introducción de datos en la parte inferior de la tabla agrega automáticamente nuevas filas. Esta es una muy buena manera de mejorar la estructura encapsulando detalles dentro de sus tablas, lo que deja a las células externas buenas posibilidades para extraer las propiedades de alto nivel como las sumas de columnas.

Utilizar tablas de pivote

Cuando tiene una tabla de Excel, y no solo entonces, es fácil usar los datos como entrada en una tabla dinámica que proporcionará la mayor parte del análisis que necesitaría en ella. Aprende a usarlo, ¡no te arrepentirás! Podría reemplazar toneladas de fórmulas de células diseñadas por el usuario y es rápido y mucho más fácil de documentar.

Extraer datos de tablas dinámicas con GetPivotData

A veces es difícil hacer que todos los datos de su tabla dinámica se confirmen al formato de informe en el que tiene que presentar sus datos comprimidos. ¡Entonces usa GetPivotData! Tiene un relleno automático de argumentos de los que puede aprender fácilmente y le permite a través de sus parámetros elegir y seleccionar de forma flexible de todos los campos visibles en sus tablas dinámicas.

Mejores prácticas

Aquí hay algunas de las mejores prácticas básicas para Excel:

- Base de datos de archivos planos: Excel **es** una aplicación de archivos planos y debe tratarse como tal
- Menos hoja de trabajo / libros de trabajo es más. El análisis será mucho más rápido con menos hojas de trabajo / libros para revisar. Trate de asegurarse de que todos los datos sin procesar estén en una hoja de trabajo en un libro de trabajo
- El diseño debe constar de 1 hoja de datos en bruto, sus datos finales provendrán de esto.
- Haga que los encabezados sean **NEGROS** esto ayuda a que Excel los reconozca como encabezados para cosas como ordenar

- Cuando coloque datos en el área de datos de su hoja de cálculo, intente evitar filas y columnas en blanco. Excel podría considerar que una fila o columna en blanco es el final de sus datos. *Sería una buena idea dejar las 4 filas superiores en blanco sobre los encabezados para usar los totales en lugar de tenerlos en la parte inferior*
- Ordene sus datos siempre que sea posible. Esto ayudará a acelerar algunas de las fórmulas y cálculos que tiene en la hoja de cálculo.
- Usa fechas reales para los encabezados y luego formatealos apropiadamente. Con esto quiero decir que si necesita nombres de mes para sus encabezados use 1/1/2017, 1/2/2017, 1/3/2017 y luego formateelos como "Mmmm". Esto es muy simple y hará la vida más fácil en el futuro cuando se usa en fórmulas.
- No ponga en 1 celda lo que puede ir en más de una. Lo que significa que si está haciendo una lista de nombres completos de usuarios, debe tener el primer nombre en una columna y el apellido en la siguiente columna (y posiblemente más columnas para los nombres medios y sufijos). Es más fácil hacer esto desde el principio que intentar hacerlo con fórmulas más adelante.
- Coloque sus encabezados en las columnas y sus datos en filas directamente debajo. Excel tiene muchas más filas que columnas. Para hacer que la hoja de cálculo sea una prueba de futuro, sus datos sin procesar deben tener los datos en las filas para que puedan continuar más allá del número de columnas disponibles.

Lea Excelentes prácticas en línea: <https://riptutorial.com/es/excel/topic/6570/excelentes-practicas>

Capítulo 7: Excelentes redondeos y precisión.

Introducción

Varias fórmulas de Excel tratan del redondeo y la precisión de los números no enteros. Esto es independiente del uso de formato de celda que afecta la visualización de datos numéricos. En algunos casos, solo usar el formato de celda es suficiente, pero en cálculos complejos, se requieren reglas estrictas de redondeo y precisión para obtener resultados consistentes y correctos.

Sintaxis

- = ROUND (número, num_digits)
- = ROUNDUP (número, num_digits)
- = REDONDEO (número, num_digits)
- = MROUND (número, múltiple)
- = TRUNC (número, [num_digits])
- = INT (número)
- = TECHO (número, significado)
- = PISO (número, significado)
- = AÚN (número)
- = ODD (número)
- = FIJO (número, [decimales], [no_commas])

Parámetros

Parámetros	Detalles
número	Número a redondear. Podría ser una celda como B2 o una constante como 3.14159
num_digits	qué lugar se redondeará 2. Omitido o 0 significa redondear a un número entero. 1 o 2 significa redondear a décimas o centésimas. -1 o -3 significa redondear a decenas o miles.
múltiple	El múltiplo al que desea redondear el número.
significado	El múltiplo al que desea redondear el número.
decimales	El número de dígitos a la derecha del punto decimal. (Opcional - por defecto a 2)
no_commas	Un valor lógico que, si es <code>TRUE</code> , evita que <code>FIXED</code> incluya comas en el texto

Parámetros	Detalles
	devuelto. (Opcional - por defecto es <code>FALSE</code>)
[...]	Los parámetros entre [corchetes] son opcionales.

Observaciones

Los valores mostrados al usuario pueden presentarse con un formato específico que no afecta a los valores de datos reales. Por ejemplo, los datos mostrados podrían formatearse como un porcentaje. Consulte [Formato de celda](#) para más detalles.

Examples

Usando la función ROUND

La función `ROUND` redondea un valor. El número de decimales para redondear se especifica mediante un valor positivo en el parámetro `num_digits` . Un valor negativo para los `num_digits` redondeará la parte entera del valor a la izquierda del punto decimal, por ejemplo, al 10 más cercano (para -1) o al 1000 más cercano (para -3).

Aquí hay una tabla que muestra cómo se puede usar la ronda.

Empezando con	RONDA (b, 2)	RONDA (b, 1)	RONDA (b)	RONDA (b, -1)
23.10651375	23.11	23.1	23	20
19.16818924	19.17	19.2	19	20
3.92748883	3.93	3.9	4	0
31.38208409	31.38	31.4	31	30
38.34235561	38.34	38.3	38	40
7.682632495	7.68	7.7	8	10
35.39315416	35.39	35.4	35	40
20.47004449	20.47	20.5	20	20
20.49775276	20.5	20.5	20	20
2.288822497	2,29	2.3	2	0

Otras funciones similares también están disponibles para controlar la dirección de redondeo:

- `ROUNDUP` : siempre redondea un número, lejos de cero.

- **ROUNDDOWN** : siempre redondea un número hacia abajo, hacia cero.

Usando las funciones TRUNC & INT

La fórmula de Excel **TRUNC** se usa para truncar un número a un número dado de lugares decimales, especificado por el parámetro `num_digits` opcional. Si este parámetro se define como un valor negativo, truncará la parte entera del valor. Si se omite el parámetro, el valor predeterminado es 0 que elimina la parte decimal del número.

La función **INT** funciona de manera similar a **TRUNC** que elimina la parte decimal de un número al redondearlo hacia abajo para dejar la parte entera. La diferencia entre los dos es cuando se realiza la operación en un número negativo; **TRUNC** eliminará el decimal, sin embargo, **INT** redondeará el valor hacia abajo desde cero.

Por ejemplo:

```
=TRUNC (123.456, 2)
=TRUNC (123.4357, -1)
=TRUNC (-123.123)
=INT (567.89)
=INT (-567.89)
```

Mostrará:

```
123.45
120.00
-123.00
567.00
-568.00
```

Usando la función MROUND

La función de Excel **MROUND** se utiliza para redondear un número a un intervalo distinto de una potencia de 10.

Estos ejemplos muestran **MROUND** al trimestre más cercano y al número par más cercano.

Empezando con	MROUND (b, 0.25)	MROUND (b, 2)
23.93195211	24.00	24
2.793135388	2,75	2
21.93903064	22.00	22
13.74193739	13.75	14
16.77047412	16.75	dieciséis
13.03922302	13.00	14

Empezando con	MROUND (b, 0.25)	MROUND (b, 2)
17.06132896	17.00	18
16.11741694	16.00	dieciséis
33.48249592	33.50	34
37.29656687	37.25	38

Se puede obtener un resultado similar utilizando las funciones `EVEN` y `ODD` que redondean un número **hasta** el número par o impar más cercano, respectivamente.

Usando las funciones `CEILING` & `FLOOR`

La función `CEILING` redondea un número, lejos de cero, al múltiplo de significación más cercano. La función `FLOOR` hace lo mismo redondeando el número hacia cero.

Un ejemplo de cuándo se puede usar `CEILING` es si desea evitar el uso de monedas en sus precios y su producto tiene un precio de \$ 4.42, use la fórmula `=CEILING(4.42, 0.05)` para redondear los precios hasta el níquel más cercano.

Por ejemplo:

```
=CEILING(2.2, 1)
=FLOOR(2.2, 1)
=CEILING(-4.8, 2)
=FLOOR(-4.8, 2)
=CEILING(0.456, 0.01)
=FLOOR(0.456, 0.01)
```

Volvería:

```
3
2
-4
-6
0.46
0.45
```

Usando la función `FIJA`

La función `FIXED` redondea un número al número especificado de decimales definido por el parámetro de `decimals`, formatea el número en formato decimal utilizando una coma como separador, a menos que se especifique como no requerido por el parámetro `no_commas`, y devuelve el resultado como texto. El parámetro `decimals` es opcional y por defecto tiene dos decimales. El parámetro `no_commas` también es opcional y por defecto es `FALSE`.

Por ejemplo:

```
=FIXED (1234.567, 1)  
=FIXED (1234.567, -1)  
=FIXED (1234.567, 1, TRUE)  
=FIXED (1234.567)
```

Volvería:

```
1,234.6  
1,230  
1234.6  
1234.57
```

Lea **Excelentes redondeos y precisión.** en línea:

<https://riptutorial.com/es/excel/topic/1871/excelentes-redondeos-y-precision->

Capítulo 8: Formato de celda

Introducción

Cada celda en una hoja de cálculo de Excel puede tener un formato único para cosas como fuente, formato de número, color, bordes y alineación. El formato condicional permite que el formato de las celdas varíe según los datos de la hoja de cálculo.

Ver también [Excel redondeo y precisión](#) .


Observaciones

El formato condicional permite que el formato del color, los efectos de fuente, el color de fondo, etc. de las celdas de la hoja de cálculo varíe según el valor de la celda u otros valores de la celda. Un rango de celdas puede tener mini-gráficos o diferentes íconos en las celdas en función de sus valores.

Examples

Formato de número

El uso más común del formato es controlar la visualización de la información numérica contenida en las celdas en un formato coherente, como la moneda, un cierto número de dígitos a la derecha del punto decimal, etc. Hay varias categorías para los números, como Moneda, Contabilidad, Porcentaje, y más. Para cada categoría, hay varias opciones disponibles:



El formato de celda *no* afecta el valor interno ni los resultados de la fórmula, solo el valor mostrado. Si una celda contiene la fórmula $=4 * \text{ACOT}(1)$ Excel puede mostrar eso como

- 3.14159265359 (Número con 11 decimales)
- \$3.14 (formato de moneda estadounidense)
- $3 \frac{1}{7}$ (Fracción con hasta un dígito)

Puede crear formatos complejos con diferentes máscaras para valores positivos, negativos y cero. Un atajo es usar la categoría existente más cercana a sus necesidades, configurar las opciones, luego hacer clic en la categoría "Personalizada", y verá los códigos de formato utilizados, y podrá modificarlos según sea necesario.

El código de formato para "Fracción con hasta un dígito" es # ?/? y si hizo clic en Personalizar e ingresó # "and" ?/? para el formato, el valor mostrado sería 3 and 1/7

Lea Formato de celda en línea: <https://riptutorial.com/es/excel/topic/9990/formato-de-celda>

Capítulo 9: Fórmulas de matriz

Examples

Suma de Producto de Rangos

En este ejemplo, el costo total de comprar alimentos se calcula al tomar el número de cada artículo, multiplicarlo por su costo y luego sumar todos esos valores.

	A	B	C	D	E
1	Item	Number	Price		
2	Apple	6	\$0.55		
3	Banana	8	\$0.25		
4	Coconut	2	\$0.89		
5	Date	20	\$0.12		
6	Eggplant	3	\$1.23		
7					
8			Total Cost:		
9			=SUM(B2:B6*C2:C6)		
10			SUM(number1, [number2], ...)		
11			\$13.17		
12					

En lugar de crear una columna separada para `Number` veces `Price` y luego sumar los valores en esa nueva columna, podemos calcular el precio total directamente mediante una fórmula de matriz:

```
=SUM(B2:B6*C2:C6)
```

Ya que esta es una fórmula matricial, debe ingresarse usando `Ctrl + Shift + Enter` para que Excel la trate como tal (de lo contrario, devolverá `#VALUE!`). Observe que si ve corchetes `{...}` alrededor de la fórmula en la barra de fórmulas, sabrá que se está evaluando como una fórmula matricial.

Así es como se evalúa esta fórmula paso a paso:

```
= SUM(B2:B6*C2:C6)
= SUM({6, 8, 2, 20, 3} * {0.55, 0.25, 0.89, 0.12, 1.23})
= SUM({6 * 0.55, 8 * 0.25, 2 * 0.89, 20 * 0.12, 3 * 1.23})
= SUM({3.30, 2.00, 1.78, 2.40, 3.69})
= 3.30 + 2.00 + 1.78 + 2.40 + 3.69
= 13.17
```

Otra forma de hacer esto es usar la función `SUMPRODUCT` :

```
=SUMPRODUCT(B2:B6,C2:C6)
```

Nota: En este caso, no es necesario utilizar `Ctrl + Shift + Enter` .

Lea Fórmulas de matriz en línea: <https://riptutorial.com/es/excel/topic/5992/formulas-de-matriz>

Capítulo 10: Función DATEDIF

Sintaxis

- = DATEDIF (fecha de inicio, fecha de finalización, unidad)

Parámetros

Unidad	Devoluciones
"Y"	El número de años completos en el período.
"METRO"	El número de meses completos en el período.
"RE"	El número de días en el período.
"MARYLAND"	La diferencia entre los días en start_date y end_date. Los meses y años de las fechas son ignorados.
"YM"	La diferencia entre los meses en start_date y end_date. Los días y años de las fechas son ignorados.
"YARDA"	La diferencia entre los días de start_date y end_date. Los años de las fechas son ignorados.

Observaciones

Tenga cuidado con los cálculos del año bisiesto cuando las unidades ignoran los años. Por ejemplo:

```
=datedif("2010-01-01","2010-07-21","YD")
```

devoluciones 201 días

```
=datedif("2016-01-01","2016-07-21","YD")
```

devuelve 202 días

Examples

Periodo de conteo entre fechas

La función `DATEDIF` devuelve la diferencia entre dos valores de fecha, según el intervalo especificado. Se proporciona para la compatibilidad con Lotus 1-2-3. La función `DATEDIF` no se

puede encontrar en la lista de funciones y las sugerencias de autocompletado y la pantalla no están disponibles. *Nota: Se pronuncia "fecha dif" en lugar de "fecha si" .*

```
=datedif("2010-01-01","2016-07-21","D")
```

Devuelve el número de días (**2393**) entre las dos fechas.

```
=datedif("2010-01-01","2016-07-21","M")
```

Devuelve el número de meses (**78**) entre las dos fechas.

```
=datedif("2010-01-01","2016-07-21","Y")
```

Devuelve el número de años (**6**) entre las dos fechas.

```
=datedif("2010-01-01","2016-07-21","MD")
```

devuelve el número de días (**20**) entre las dos fechas, ignorando los meses y años

```
=datedif("2010-01-01","2016-07-21","YM")
```

devuelve el número de meses (**6**) entre las dos fechas, ignorando los años

```
=datedif("2010-01-01","2016-07-21","YD")
```

devuelve el número de días (**201**) entre las dos fechas, ignorando los años

Lea Función DATEDIF en línea: <https://riptutorial.com/es/excel/topic/2786/funcion-datedif>

Capítulo 11: Función MATCH

Introducción

(Opcional) Cada tema tiene un enfoque. Dígales a los lectores lo que encontrarán aquí y hágales saber a los futuros colaboradores lo que les pertenece.

Parámetros

Parámetro	Descripción
valor de búsqueda	El valor que desea igualar. Puede ser un valor fijo, una referencia de celda o un rango con nombre. Las cadenas no pueden exceder los 255 caracteres (<i>requerido</i>)
lookup_array	La referencia de celda (o rango con nombre) que desea buscar, puede ser una fila o una columna ordenada en orden ascendente para coincidencias de tipo 1 por defecto; orden descendente para coincidencias tipo -1; o cualquier orden para coincidencias de tipo 0 (<i>requerido</i>)
tipo de concordancia	Controla la forma en que funciona la búsqueda. Establézcalo en 0 si solo quiere coincidencias exactas, establezca en 1 si desea hacer coincidir elementos menores o iguales a su valor de búsqueda , o -1 si desea hacer coincidir elementos mayores o iguales a su valor de búsqueda . (<i>Opcional</i> - por defecto a 1)

Observaciones

Propósito

Utilice la función MATCH para verificar si (y dónde) se puede encontrar un valor en una lista. A menudo se ve como un retorno de parámetro para la fila y / o columna en la función INDEX (matriz, fila, columna). Permite referencias de fila / columna negativas que permiten búsquedas a la izquierda o arriba.

Funciones similares:

- **VLOOKUP** : como MATCH, pero devuelve datos de la tabla, en lugar del número de fila o columna. Solo puede buscar una tabla verticalmente y devolver valores en o a la derecha del valor encontrado.
- **HLOOKUP**: como MATCH, pero devuelve datos de la tabla, en lugar del número de fila o columna. Solo se puede buscar en una tabla horizontalmente y devolver valores en o por debajo del valor encontrado.

Examples

Comprobando si una dirección de correo electrónico aparece en una lista de direcciones

Digamos que necesita verificar si una dirección de correo electrónico aparece en una larga lista de direcciones de correo electrónico.

	A	B	C	D	E	F
1	ID	GivenName	Surname	EmailAddress		Emails I'm looking for
2	1	Daniel	Cortez	Daniel.A.Cortez@dodgit.com		Allan.J.Morgan@mailinator.com
3	2	Charles	Russ	Charles.R.Russ@dodgit.com		Charles.R.Russ@dodgit.com
4	3	Victor	Platt	Victor.T.Platt@pookmail.com		Daniel.A.Cortez@dodgit.com
5	4	George	Lehman	George.T.Lehman@mailinator.com		David.D.Spencer@trashymail.com
6	5	Katie	Griffith	Katie.M.Griffith@trashymail.com		Deborah.P.Myers@spambob.com
7	6	Deborah	Myers	Deborah.P.Myers@spambob.com		George.T.Lehman@mailinator.com
8	7	Jennifer	Romano	Jennifer.R.Romano@pookmail.com		Gerald.M.Parker@mailinator.com
9	8	Allan	Morgan	Allan.J.Morgan@mailinator.com		Jamie.M.Johnson@mailinator.com
10	9	Mary	Rice	Mary.J.Rice@trashymail.com		Janice.J.Joachim@dodgit.com
11	10	Patsy	Chafin	Patsy.J.Chafin@trashymail.com		Jennifer.R.Romano@pookmail.com
12	11	Laurence	Marano	Laurence.A.Marano@spambob.com		John.T.Beasley@trashymail.com
13	12	John	Beasley	John.T.Beasley@trashymail.com		Johnny.R.Desjardins@dodgit.com
14	13	John	Mejia	John.E.Mejia@pookmail.com		Katie.M.Griffith@trashymail.com
15	14	Daniel	Torres	Daniel.G.Torres@mailinator.com		Laurence.A.Marano@spambob.com
16	15	Stephanie	Cardenas	Stephanie.E.Cardenas@spambob.com		Mary.J.Rice@trashymail.com
17	16	Donald	Bowen	Donald.K.Bowen@dodgit.com		Michelle.C.Gonzales@dodgit.com
18	17	Gloria	Arrowood	Gloria.W.Arrowood@mailinator.com		Nicholas.J.Melvin@trashymail.com
19	18	Roslyn	Mahaney	Roslyn.O.Mahaney@pookmail.com		Patsy.J.Chafin@trashymail.com
20	19	Lois	Smith	Lois.G.Smith@mailinator.com		Victor.T.Platt@pookmail.com
21	20	James	Ward	James.E.Ward@pookmail.com		Yvonne.H.Neff@spambob.com
22	21	Frances	Jones	Frances.R.Jones@spambob.com		
23	22	Yvette	Lowery	Yvette.A.Lowery@dodgit.com		

Utilice la función MATCH para devolver el número de fila en el que se puede encontrar la dirección de correo electrónico. Si no hay coincidencia, la función devuelve un error # N / A.

```
=MATCH(F2, $D$2:$D$200, 0)
```

- El valor del que está recuperando datos está en la celda **F2**.
- El rango que está buscando se encuentra en **\$ D \$ 2 : \$ D \$ 200**
- Solo quieres saber dónde hay una coincidencia exacta (**0**)

Pero es posible que no le importe en qué número de fila está la dirección de correo electrónico; solo desea saber si existe, de modo que podamos ajustar la función MATCH para devolver *Sí* o *Falta* :

```
=IFERROR(IF(MATCH(F2, $D$2:$D$200, 0), "Yes"), "Missing")
```

Combinando MATCH con INDEX

Digamos que tienes un conjunto de datos que consta de nombres y direcciones de correo electrónico. Ahora en otro conjunto de datos, solo tiene la dirección de correo electrónico y desea encontrar el nombre apropiado que pertenece a esa dirección de correo electrónico.

	A	B	C	D	E	F
1	ID	Name	Surname	Email		Email
2	1	oona	hietala	oona.hietala@example.com		alexander.baier@example.com
3	2	karlisa	almeida	karlisa.almeida@example.com		maria.fischer@example.com
4	3	alisa	lassila	alisa.lassila@example.com		morgan.francois@example.com
5	4	maria	fischer	maria.fischer@example.com		camille.guillot@example.com
6	5	halit	van geere	halit.vangeerenstein@example.com		louanne.mercier@example.com
7	6	morgan	francois	morgan.francois@example.com		alex.bell@example.com
8	7	lucas	walker	lucas.walker@example.com		oona.hietala@example.com
9	8	ross	watts	ross.watts@example.com		ross.watts@example.com
10	9	alex	bell	alex.bell@example.com		selma.johansen@example.com
11	10	laura	rasmusser	laura.rasmussen@example.com		alex.bell@example.com
12	11	gerardus	schelhaas	gerardus.schelhaas@example.com		
13	12	selma	johansen	selma.johansen@example.com		
14	13	vanessa	maier	vanessa.maier@example.com		
15	14	alexander	baier	alexander.baier@example.com		
16	15	louanne	mercier	louanne.mercier@example.com		
17	16	tiago	dupuis	tiago.dupuis@example.com		
18	17	ellen	ellis	ellen.ellis@example.com		
19	18	alicia	harris	alicia.harris@example.com		
20	19	camille	guillot	camille.guillot@example.com		
21	20	jason	cruz	jason.cruz@example.com		
22	21	lester	long	lester.long@example.com		
23	22	julia	burke	julia.burke@example.com		
24	23	arthur	nguyen	arthur.nguyen@example.com		
25	24	theo	bischoff	theo.bischoff@example.com		
26	25	meral	yilmazer	meral.yilmazer@example.com		
27						

La función MATCH devuelve la fila correspondiente a la que se encuentra el correo electrónico y la función INDEX lo selecciona. Del mismo modo, esto también se puede hacer para las columnas. Cuando no se puede encontrar un valor, devolverá un error **# N / A**.

Este es un comportamiento muy similar a VLOOKUP O HLOOKUP, pero es mucho más rápido y combina ambas funciones anteriores en una.

- Busque el valor de la celda **F2 (alexander.baier@example.com)**
- Dentro del conjunto de datos **\$ D \$ 2: \$ D \$ 26**
- Utilizar coincidencia exacta (**0**)
- Utilice el número de la fila resultante (**14**) de un conjunto de datos diferente **\$ B \$ 2: \$ B \$ 26**

Lea Función MATCH en línea: <https://riptutorial.com/es/excel/topic/4419/funcion-match>

Capítulo 12: Función SUMPRODUCT

Introducción

La función SUMPRODUCT multiplica los componentes correspondientes en los arreglos dados y devuelve la suma de esos productos.

Sintaxis

- SUMPRODUCT (array1, [array2], [array3], ...)

Observaciones

- Los argumentos de la matriz deben tener las mismas dimensiones. Si no lo hacen, SUMPRODUCT devuelve el #VALUE! valor de error.
- SUMPRODUCT trata las entradas de la matriz que no son numéricas como si fueran ceros.

Examples

Usando SUMPRODUCT con rangos numéricos

Considere los rangos $A1:A3$ y $B1:B3$ tienen el mismo tamaño y solo valores numéricos, como se muestra a continuación

	A	B
1	1	4
2	2	5
3	3	6

```
=SUMPRODUCT(A1:A3, B1:B3)
```

Esto recorrerá los rangos, tomando el producto de los valores en la misma fila y sumándolos, obteniendo 32 en este ejemplo.

```
A1*B1 = 4  
A2*B2 = 10  
A3*B3 = 18
```

Usando SUMPRODUCT con matrices booleanas

Considere los siguientes rangos $A1:A3$ y $B1:B3$ siguiente manera

	A	B
1	a	4
2	b	5
3	c	6

```
=SUMPRODUCT(--(A1:A3="c"),B1:B3)
```

Esto primero manipulará (A1:A3="c") en la siguiente matriz

```
A1="c" = FALSE
A2="c" = FALSE
A3="c" = TRUE
```

Luego aplique el operador -- que convierte TRUE y FALSE en 1 y 0 , respectivamente. Así que la matriz se convierte

```
--FALSE = 0
--FALSE = 0
--TRUE = 1
```

Luego, la fórmula SUMPRODUCT se completa como en el caso numérico simple. Volviendo 6 en este ejemplo

```
0*4 = 0
0*5 = 0
1*6 = 6
```

Nota: este es el equivalente de una función SUMIF

Lea Función SUMPRODUCT en línea: <https://riptutorial.com/es/excel/topic/8096/funcion-sumproduct>

Capítulo 13: Índice de coincidencia para Excel

Introducción

Una alternativa más versátil a VLOOKUP. Una coincidencia de índice reúne el poder de Vlookup y Hlookup en una fórmula. Tampoco necesita saber qué número de columna / fila es la información. Debido a esto, eliminar columnas / filas no estropeará la fórmula.

Examples

Índice de coincidencia vertical

Índice de coincidencia vertical

	A	B	C	D	E	F	G	H
1	Apple	Red	Fruit		Find			
2	Grape	Purple	Fruit		Vegetable	=index(A1:A3,match(E2,C1:C3,0))		
3	Cucumber	Green	Vegetable					

```
=INDEX(A1:A3, MATCH(E2, C1:C3, 0))
```

Coincidencia de índice horizontal

Coincidencia de índice horizontal

	A	B	C	D	E	F	G	H
1	Apple	Grape	Cucumber		Find			
2	Red	Purple	Green		Vegetable	=index(A1:C1,match(E2,A3:C3,0))		
3	Fruit	Fruit	Vegetable					

```
=INDEX(A1:C1, MATCH(E2, A3:C3, 0))
```

Lea Índice de coincidencia para Excel en línea: <https://riptutorial.com/es/excel/topic/9313/indice-de-coincidencia-para-excel>

Capítulo 14: Un proceso fácil para convertir datos mensuales a trimestrales en Excel

Introducción

Un deseo frecuente es convertir los datos mensuales en formato de datos trimestrales. Un método simple es 1) sumar los meses apropiados en sumas trimestrales, luego 2) identificar esos meses como trimestres, y finalmente 3) filtrar los datos trimestrales de sus datos mensuales. Aquí hay un método relativamente rápido y fácil ilustrado a través del siguiente ejemplo:

Examples

Una forma fácil de realizar una tarea rutinaria a veces tediosa:

Si tiene sus datos con cada mes de datos ordenados en filas, así:

	A	B	C
1	Date	Product A	Service A
2	2000-01-31	178.00	358.00
3	2000-02-29	619.00	553.00
4	2000-03-31	152.00	106.00
5	2000-04-30	662.00	102.00
6	2000-05-31	230.00	290.00
7	2000-06-30	248.00	583.00
8	2000-07-31	438.00	413.00
9	2000-08-31	428.00	401.00
10	2000-09-30	481.00	488.00
11	2000-10-31	631.00	433.00
12	2000-11-30	436.00	449.00
13	2000-12-31	378.00	300.00
14	2001-01-31	641.00	478.00
15	2001-02-28	560.00	550.00
16	2001-03-31	310.00	534.00
17	2001-04-30	301.00	402.00
18	2001-05-31	631.00	397.00
19	2001-06-30	490.00	268.00
20	2001-07-31	234.00	378.00
21	2001-08-31	516.00	220.00
22	2001-09-30	167.00	116.00

Comience por crear sumas trimestrales en las columnas adyacentes, D y E en nuestro ejemplo.

Comience con la tercera fila en la nueva columna o el primer trimestre que desea crear, en este ejemplo usaremos el 31 de marzo (2000-03-31). Utilice la función de resumen durante los dos meses anteriores y el trimestre, para el uso de marzo:

= SUMA (B2: B4)

Luego use una función similar para el total en la siguiente columna E. Tendrá la siguiente fórmula:

= SUMA (C2: C4)

No se preocupe de que las sumas se crean para los meses que no son trimestres. Esto se hace para que la copia sea más fácil para usted. No necesitarás esos valores de todos modos en los siguientes pasos.

Finalmente, necesitamos identificar los cuartos sin mucho trabajo. Si su columna Fecha no está en un formato de Fecha de Excel, como texto, es posible que desee convertirla. En una nueva columna, F en este ejemplo, comience esta fórmula en la primera fila de datos. Marcaremos las filas usando la siguiente fórmula:

= IF (MOD (MES (A2), 3) = 0, "Trimestre", "Mes")


Ahora copie las tres fórmulas en las nuevas columnas D, E y F en las celdas restantes de su tabla. Debe tener un aspecto como este:

	A	B	C	D	E	F
1	Date	Product A	Service A	Qtr Prod A	Qtr Serv A	Period
2	2000-01-31	330.00	621.00			Month
3	2000-02-29	245.00	664.00			Month
4	2000-03-31	370.00	230.00	945.00	1,515.00	Quarter
5	2000-04-30	437.00	467.00	1,052.00	1,361.00	Month
6	2000-05-31	629.00	260.00	1,436.00	957.00	Month
7	2000-06-30	218.00	198.00	1,284.00	925.00	Quarter
8	2000-07-31	214.00	661.00	1,061.00	1,119.00	Month
9	2000-08-31	309.00	136.00	741.00	995.00	Month
10	2000-09-30	239.00	566.00	762.00	1,363.00	Quarter
11	2000-10-31	126.00	103.00	674.00	805.00	Month
12	2000-11-30	201.00	375.00	566.00	1,044.00	Month
13	2000-12-31	587.00	672.00	914.00	1,150.00	Quarter
14	2001-01-31	341.00	376.00	1,129.00	1,423.00	Month
15	2001-02-28	484.00	186.00	1,412.00	1,234.00	Month
16	2001-03-31	535.00	141.00	1,360.00	703.00	Quarter
17	2001-04-30	594.00	210.00	1,613.00	537.00	Month
18	2001-05-31	309.00	246.00	1,438.00	597.00	Month
19	2001-06-30	514.00	145.00	1,417.00	601.00	Quarter
20	2001-07-31	665.00	568.00	1,488.00	959.00	Month
21	2001-08-31	408.00	666.00	1,587.00	1,379.00	Month
22	2001-09-30	456.00	233.00	1,529.00	1,467.00	Quarter

Finalmente, seleccione esta tabla, elija **Formato como tabla** , elija un formato simple y Excel producirá un resultado como este:

	A	B	C	D	E	F
1	Date	Product A	Service A	Qtr Prod A	Qtr Serv A	Period
2	2000-01-31	343.00	594.00			Month
3	2000-02-29	278.00	425.00			Month
4	2000-03-31	286.00	175.00	907.00	1,194.00	Quarter
5	2000-04-30	548.00	403.00	1,112.00	1,003.00	Month
6	2000-05-31	452.00	305.00	1,286.00	883.00	Month
7	2000-06-30	226.00	257.00	1,226.00	965.00	Quarter
8	2000-07-31	393.00	546.00	1,071.00	1,108.00	Month
9	2000-08-31	622.00	185.00	1,241.00	988.00	Month
10	2000-09-30	104.00	576.00	1,119.00	1,307.00	Quarter
11	2000-10-31	517.00	623.00	1,243.00	1,384.00	Month
12	2000-11-30	101.00	542.00	722.00	1,741.00	Month
13	2000-12-31	164.00	192.00	782.00	1,357.00	Quarter
14	2001-01-31	130.00	328.00	395.00	1,062.00	Month
15	2001-02-28	393.00	559.00	687.00	1,079.00	Month
16	2001-03-31	238.00	183.00	761.00	1,070.00	Quarter
17	2001-04-30	516.00	485.00	1,147.00	1,227.00	Month
18	2001-05-31	313.00	289.00	1,067.00	957.00	Month
19	2001-06-30	316.00	422.00	1,145.00	1,196.00	Quarter
20	2001-07-31	232.00	416.00	861.00	1,127.00	Month
21	2001-08-31	612.00	175.00	1,160.00	1,013.00	Month
22	2001-09-30	167.00	277.00	1,011.00	868.00	Quarter

El último paso es seleccionar el **botón de filtro** sobre la nueva columna "Período" en F y seleccionar solo los valores de "Cuartos" para filtrar:



Tu mesa debe tener este aspecto ahora:

	A	B	C	D	E	F
1	Date	Product A	Service A	Qtr Prod A	Qtr Serv A	Period
4	2000-03-31	343.00	630.00	1,375.00	1,234.00	Quarter
7	2000-06-30	148.00	654.00	1,042.00	1,043.00	Quarter
10	2000-09-30	385.00	161.00	1,618.00	813.00	Quarter
13	2000-12-31	118.00	477.00	1,257.00	1,525.00	Quarter
16	2001-03-31	602.00	440.00	1,308.00	1,150.00	Quarter
19	2001-06-30	308.00	420.00	784.00	897.00	Quarter
22	2001-09-30	479.00	433.00	1,466.00	1,580.00	Quarter

En este punto, sugeriría seleccionar solo las columnas que necesita de su tabla final y luego usar **Pegar-> Especial-> Valores** para reubicar sus datos.

Lea [Un proceso fácil para convertir datos mensuales a trimestrales en Excel en línea:](https://riptutorial.com/es/excel/topic/10837/un-proceso-facil-para-convertir-datos-mensuales-a-trimestrales-en-excel)
<https://riptutorial.com/es/excel/topic/10837/un-proceso-facil-para-convertir-datos-mensuales-a-trimestrales-en-excel>

Capítulo 15: VLOOKUP

Introducción

Busca un valor en la primera columna de una matriz de tabla y devuelve un valor en la misma fila de otra columna de la matriz de tabla.

La V en VLOOKUP significa vertical. Use VLOOKUP en lugar de HLOOKUP cuando sus valores de comparación estén ubicados en una columna a la izquierda de los datos que desea encontrar.

Sintaxis

- VLOOKUP (lookup_value, table_array, col_index_num, range_lookup)

Parámetros

Parámetro	Descripción
valor de búsqueda	El valor que está buscando en la columna izquierda de la tabla. Puede ser un valor fijo, una referencia de celda o un rango con nombre (<i>requerido</i>)
matriz de tabla	El rango de celdas que consta de la columna en la que desea buscar en el lado izquierdo con valores en celdas a la derecha que desea que se devuelvan. Puede ser una referencia de celda de Excel o un rango con nombre. (<i>requerido</i>)
col_index_num	El número de la columna de la que desea devolver datos, contando desde la columna más a la izquierda de su tabla (<i>requerido</i>)
rango de búsqueda	Controla la forma en que funciona la búsqueda. Si FALSO o 0, Excel realizará una búsqueda exacta y regresará solo donde haya una coincidencia exacta en la columna de la izquierda. La atención a la precisión con el redondeo es muy importante para esta búsqueda con valores numéricos. Si es VERDADERO o 1, Excel realizará y aproximará la búsqueda y devolverá el último valor que iguala o supera. Como tal, la primera columna debe ordenarse en orden ascendente y una búsqueda aproximada. range_lookup . (<i>Opcional</i> - por defecto es TRUE)

Observaciones

Funciones similares:

- HLOOKUP (igual que VLOOKUP pero busca horizontalmente en lugar de verticalmente)

- **MATCH** (si su valor de búsqueda tiene una coincidencia, devuelve el número de fila dentro del rango)
- **LOOKUP** (similar a VLOOKUP y MATCH, y se proporciona para compatibilidad con versiones anteriores)

Errores comunes:

- No establecer el parámetro **range_lookup** y obtener el comportamiento de coincidencia no exacto predeterminado
- Sin **corrección** y rango de direcciones absolutas en la **matriz de tabla** : al copiar una fórmula, la referencia de "tabla de búsqueda" también se mueve

Examples

Usar VLOOKUP para obtener el apellido de una persona de su ID de empleado

Vlookup encuentra algún valor en la columna de la *izquierda* de un rango y devuelve un valor un cierto número de columnas a la derecha y en la misma fila.

Supongamos que desea buscar el apellido de Employee ID 2 en esta tabla:

	A	B	C
1	EmpID	First name	Surname
2	1	Joe	Bloggs
3	2	Linda	Williams
4	3	John	Smith
5			

```
=VLOOKUP (2, $A$2:$C$4, 3, 0)
```

- El valor por el que está recuperando datos es **2**.
- La tabla que está buscando se encuentra en el rango **\$ A \$ 2: \$ C \$ 4**
- La columna de la que desea devolver datos es la **tercera** columna de la izquierda
- Solo desea devolver resultados donde haya una coincidencia exacta (**0**)

Tenga en cuenta que si no hay una coincidencia exacta en el ID de empleado, **VLOOKUP** devolverá #N/A

Usar VLOOKUP para calcular el porcentaje de bonificación (ejemplo con el comportamiento "predeterminado")

En la mayoría de los casos, el **range_lookup** se utiliza como FALSE (una coincidencia exacta). El valor predeterminado para este parámetro es VERDADERO: se usa con menos frecuencia en este formulario, pero este ejemplo muestra un caso de uso. Un supermercado da un bono basado en el gasto mensual de los clientes.

	A	B
	Monthly spend	
1	(on or over) ▾	Bonus ▾
2	0	0%
3	250	1%
4	500	2.5%
5	750	4%
6	1000	5%
7		

Si el cliente gasta 250 EUR o más en un mes, obtiene un bono del 1%; 500 EUR o más da 2.5%, etc. Por supuesto, ¡el cliente no siempre gastará exactamente uno de los valores en la tabla!

```
=VLOOKUP(261, $A$2:$B$6, 2, TRUE)
```

En este modo, VLOOKUP encontrará el primer valor en la columna A (siguiendo los pasos ascendentes) **menor o igual al** valor 261 -> es decir, obtendremos el valor del 1% devuelto. **Para esta coincidencia no exacta, la tabla debe ordenarse en orden ascendente de la primera columna.**

- El valor del que está recuperando datos es de **261**.
- La tabla que está buscando se encuentra en el rango **\$ A \$ 2: \$ B \$ 6**
- La columna que desea devolver datos de la columna es el **2º** desde la izquierda
- Solo desea devolver los resultados donde hay una coincidencia no exacta (**VERDADERO**) *que podríamos omitir esta VERDADERO ya que es el valor predeterminado*

Utilizando VLOOKUP con coincidencia aproximada.

Cuando se **omite el** parámetro **range_lookup** , TRUE o 1, VLOOKUP encontrará una coincidencia aproximada. Por "aproximado", queremos decir que VLOOKUP coincidirá con el valor más pequeño que sea mayor que su valor de **búsqueda** . Tenga en cuenta que su **table_array** *debe* ordenarse en orden ascendente por valores de búsqueda. Los resultados serán impredecibles si sus valores no están ordenados.

	A	B	C	D	E	F
1	ID	Value		Search ID	Formula	Result
2	1	B		4	=VLOOKUP(D2,A\$2:B\$7,2,TRUE)	C
3	3	C		4.5	=VLOOKUP(D3,A\$2:B\$7,2,TRUE)	C
4	5	D		5	=VLOOKUP(D4,A\$2:B\$7,2,TRUE)	D
5	7	E		5.5	=VLOOKUP(D5,A\$2:B\$7,2,TRUE)	D
6	9	F		6	=VLOOKUP(D6,A\$2:B\$7,2,TRUE)	D
7	11	G		6.5	=VLOOKUP(D7,A\$2:B\$7,2,TRUE)	D

Usando VLOOKUP con coincidencia exacta

La idea central de `VLOOKUP` es buscar información en una tabla de hoja de cálculo y colocarla en otra.

Por ejemplo, supongamos que esta es la tabla en la Hoja1:

John	12/25/1990
Jane	1/1/2000

En la Hoja 2, coloque a `John` , `Andy` y `Jane` en A1, A2 y A3.

En B1, a la derecha de `John` , coloqué:

```
=VLOOKUP (A1, Sheet1!$A$1:$B$4, 2, FALSE)
```

Aquí hay una breve explicación de los parámetros dados a VLOOKUP. La A1 significa que estoy buscando a `John` en la A1 de la Hoja2. los

```
Sheet1!$A$1:$B$4
```

le dice a la función que mire la Hoja 1, las columnas A a B (y las filas 1 a 4). Los signos de dólar son necesarios para que Excel use referencias absolutas (en lugar de relativas). (Las referencias relativas harían que todo cambiara de manera indeseable al copiar la fórmula).

El `2` significa devolver la segunda columna, que es la fecha.

El `FALSE` significa que está requiriendo una coincidencia exacta.

Luego copié B1 a B2 y B3. (La forma más fácil de hacer esto sería hacer clic en B1 para resaltarlo. Luego, mantenga presionada la tecla Mayús y presione la flecha hacia abajo dos veces. Ahora B1, B2 y B3 están resaltados. Luego presione Ctrl-D para llenar la fórmula . Si se hace correctamente, uno debería tener la misma fórmula en B3 que en B1. *Si las fórmulas para la tabla de búsqueda están cambiando* , por ejemplo, de la `Sheet1!A1:B3` a la `Sheet1:A3:B5` , entonces debe usar referencias absolutas (con signos de dólar) para evitar el cambio.

Aquí están los resultados:

John	12/25/1990
Andy	#N/A
Jane	1/1/2000

Encontró a John y Jane, y les devolvió sus fechas de nacimiento. No encontró a Andy, por lo que muestra un `#N/A`

Lea VLOOKUP en línea: <https://riptutorial.com/es/excel/topic/4327/vlookup>

Creditos

S. No	Capítulos	Contributors
1	Empezando con Excel	4444 , Andre Terra , Community , dot.Py , Mike , T.Furholzer , Washington Guedes
2	Contando celdas únicas	Ulli Schmid
3	Conversión de base	SeanC
4	Crear conexión a otros archivos de Excel	petergensler
5	Excelencias y límites	paul bica
6	Excelentes prácticas	Alon Eitan , Mats Lind , Mike
7	Excelentes redondeos y precisión.	Mark Stewart , mike7mike , nekomatic , rajah9
8	Formato de celda	Mark Stewart
9	Fórmulas de matriz	Alexis Olson
10	Función DATEDIF	Andi Mohr , CallumDA , dav , Islam Tawfik
11	Función MATCH	Andi Mohr , japborst , Mark Fitzgerald
12	Función SUMPRODUCT	CallumDA
13	Índice de coincidencia para Excel	Mukul215 , Washington Guedes
14	Un proceso fácil para convertir datos mensuales a trimestrales en Excel	Klatuu
15	VLOOKUP	Andi Mohr , Captain , EBH , Forward Ed , japborst , picobit , rajah9 , Shrikant , Washington Guedes